
1 Aufgabenstellung 
In diesem Beitrag wird beschrieben, wie ein einmaliger Termin im OUTLOOK-Kalender angelegt 

werden kann, ohne mit den evtl. schon vorhandenen Terminen des relevanten Tages zu kollidieren. 

Zur Ermittlung der/des verfügbaren Zeitfenster(s) wird ein Benutzerformular eingesetzt, das auf 

folgenden Annahmen beruht: Ein verfügbares Zeitfenster muss … 

 größer oder gleich der normalen Dauer (in ganzen Minuten) eines Termins sein, 

 zwischen dem normalen Beginn und Ende der Arbeitszeit eines Tages liegen. 

Wenn ein oder mehr verfügbare Zeitfenster gefunden werden, kann der Benutzer des Formulars 
auswählen, welches davon zur automatischen Erstellung eines neuen Einzeltermins im OUTLOOK-
Kalender verwandt werden soll. 

2 Ausgangssituation 
Die angenommene Ausgangssituation ist in Abbildung 1 dargestellt. Sie beinhaltet drei vorhandene 

Einzeltermine für den 22.10.2019: 

 

Abbildung 1: Drei vorhandene Einzeltermine am 22.10.2019 

Wenn ein Termin jeweils 30 Minuten dauert und wenn die normale Arbeitszeit am genannten Tag 

um 08.30 Uhr beginnt und um 19:00 Uhr endet, dann existieren 4 verfügbare Zeitfenster zur 

automatischen Erstellung eines neuen Einzeltermins (s. Tabelle 1): 

Zeitfenster 

Beginn Ende 

08:30 09:30 

10:00 13:00 

13:30 16:00 

16:30 19:00 

Tabelle 1: Vier verfügbare Zeitfenster am 22.10.2019 

3 Schrittfolge im Benutzerformular 
Das im Folgenden präsentierte Benutzerformular (s. Abbildung 2) umfasst 3 Schritte: 

1. Verfügbare Zeitfenster für einen Einzeltermin an einem bestimmten Datum ermitteln. 

2. Eines davon auswählen. 

3. Mit dem gewählten Zeitfenster einen Einzeltermin im OUTLOOK-Kalender erstellen. 

3.1 Erster Schritt: Verfügbare Zeitfenster finden 
Zur Bestimmung der verfügbaren Zeitfenster sind vier Parameter erforderlich (s. Abbildung 2, 

Rahmen 'Parameter …'). Diese Parameter sind änderbar. Bei Änderung wird ihre Plausibilität durch 

programmierte Kontrollen1 sichergestellt. 

 

Abbildung 2: Vorbelegtes Benutzerformular zum Finden verfügbarer Zeitfenster 

Die Formularfelder im Rahmen 'Eigenschaften des einmaligen Termins …' sind während der Suche 

nach evtl. verfügbaren Zeitfenstern deaktiviert (erkennbar an der jeweils grauen Hintergrundfarbe). 

Durch Betätigung der Schaltfläche 'Verfügbare Zeitfenster finden' (s. Abbildung 2) werden diese mit 

einer VBA-Funktion ermittelt und in dem Kombinationsfeld 2  mit der Beschriftung 'Bitte ein 

verfügbares Zeitfenster wählen' angezeigt (s. Abbildung 3). 

                                                             
1 Durch sog. Exit-Ereignisprozeduren (s. Tabelle 4). 
2  Dieses Steuerelement ist eine Mischung aus einem Text- und einem Listenfeld, ohne dafür viel Platz im 

Benutzerformular zu beanspruchen. 

3.2 Zweiter Schritt: Verfügbares Zeitfenster auswählen 
Der Benutzer kann nun wählen, ob er ein verfügbares Zeitfenster akzeptiert und anschließend mit 

der Schaltfläche 'Einmaligen Termin erstellen' einen solchen Termin automatisch generiert oder die 

VBA-Prozedur mit der Schaltfläche 'Abbruch/Ende' sofort verlässt. 

 

Abbildung 3: Benutzerformular mit verfügbaren Zeitfenstern 

3.3 Dritter Schritt: Einmaligen Termin erstellen 
Bevor der Benutzer die Schaltfläche 'Einmaligen Termin erstellen' betätigt, hat er die Möglichkeit, vier 

Vorschlagswerte im Rahmen ‚Eigenschaften des einmaligen Termins‘ zu ändern: 

 Betreff 

 Ort 

 Erinnerung 

 Textkörper 

Die grau eingefärbten Formularfelder (Start am/um, Dauer) sind aus logischen Gründen nicht 

änderbar. 

Bei Betätigung der oben genannten Schaltfläche, wird eine Erfolgsmeldung (s. Abbildung 4) angezeigt 

und ein entsprechender Termin mit dem Betreff 'Neuer Termin' im OUTLOOK-Kalender angelegt (s. 

Abbildung 5 und Abbildung 6). 

Die vier Textfelder im Rahmen 'Parameter …' sind bei Erstellung eines einmaligen Termins aus 

logischen Gründen deaktiviert (d.h. die jeweilige Hintergrundfarbe ist grau, s. Abbildung 3). 

 

Abbildung 4: Erfolgsmeldung nach Erstellung eines einmaligen Termins 

 

Abbildung 5: 'Neuer Termin' in OUTLOOK-Kalender am 22.10.2019 um 10:00 Uhr 

 

Abbildung 6: Drei vorhandene Einzeltermine und ein 'Neuer Termin' im OUTLOOK-Kalender am 22.10.2019 

4 Gestaltung des Benutzerformulars 

4.1 Steuerelemente 
Das schon mehrmals gezeigte Benutzerformular mit dem Namen 'frmZeitfenster' enthält 51 

Steuerelemente3 (s. Tabelle 2). Alle Steuerelemente sind mit einem Präfix versehen, der ihren 

jeweiligen Typ eindeutig angibt. 

Steuerelemente des Benutzerformulars 

Typ Präfix Anzahl 
Befehlsschaltfläche cmd 3 

Bezeichnungsfeld lbl 28 

Kombinationsfeld cbo 1 

Kontrollkästchen chk 1 

Rahmen fra 4 

Textfeld txt 14 

Summe 51 

Tabelle 2: Steuerelemente des Benutzerformulars in alphabetischer Reihenfolge 

4.2 Einzelheiten 
Tabelle 3 enthält die Variablennamen des Benutzerformulars. Sie dienen zum Verständnis des 

dazugehörigen VBA-Quellcodes (s. Listing 1): 

Feldbezeichnung  Feldtyp Variablenname 

... im Rahmen 'Parameter' 

Datum Textfeld txtDate 

Normale Dauer eines Termins Textfeld txtDuration 

Frühester Beginn eines Termins Textfeld txtStartTime 

Spätestes Ende eines Termins Textfeld txtEndTime 

(Format: dd.mm.yyy) Bezeichnungsfeld  

(in Minuten) Bezeichnungsfeld  

(Format hh:nn) Bezeichnungsfeld  

(Format hh:nn) Bezeichnungsfeld  

... im Rahmen ‚Verfügbarkeit‘ 

Bitte ein verfügbares Zeitfenster wählen Kombinationsfeld cboTimeSlots 

... im Rahmen 'Eigenschaften des einmaligen Termins' 

Betreff Textfeld txtTopic 

Ort Textfeld txtWhere 

Start am/um Textfeld txtBegin 

Dauer Textfeld txtLength 

Erinnerung Kontollkästchen chkRemind 

Textkörper Textfeld txtComment 

(nicht änderbar, s. gewähltes Zeitfenster) Bezeichnungsfeld  

(in Minuten, nicht änderbar, siehe oben) Bezeichnungsfeld  

Ja/Nein Bezeichnungsfeld  

Sonstige Formularfelder 

Verfügbare Zeitfenster finden Befehlsschaltfläche cmdFind 

Einmaligen Termin erstellen Befehlsschaltfläche cmdCreate 

Abbruch/Ende Befehlsschaltfläche cmdCancel 

Tabelle 3: Variablennamen für das Benutzerformular 

                                                             
3 Direktfenster: ? Userform2.Controls.Count ergibt 51 

5 Quellcode der Anwendung 

5.1 Prozedurtypen 
Im VBA-Quellcode der Anwendung befinden sich 17 Private4 Prozeduren (s. Tabelle 4): 

VBA-Prozeduren 

Anzahl Typ 

1 AfterUpdate-Ereignisprozedur 

1 Change-Ereignisprozedur 

3 Click-Ereignisprozedur 

6 Exit-Ereignisprozedur 

4 Function-Prozedur 

5 Userform-Prozedur 

 Tabelle 4: Prozedurtypen im Quellcode der Anwendung 

5.2 Listing mit Zeilennummern 
Das Listing des gesamten VBA-Quellcodes der Anwendung umfasst 355 Befehlszeilen (s. Listing 1). Im 

Listing ist der jeweilige Beginn der 17 Prozeduren (s. Tabelle 4) gelb hervorgeben. Die Hintergrund-

farbe der sog. Inline-Kommentare ist grau. 

Kern der Anwendung ist die Funktionsprozedur mit dem Namen 'FindAvailableTimeSlots' (s. Listing 1, 

Zeile 50 – 140). Sie ermittelt verfügbare Zeitfenster im OUTLOOK-Kalender an einem vorgegebenen 

Datum. 

6 Erweiterung der Anwendung für mehrere Benutzer 
Die weiter oben beschriebene Anwendung zur Ermittlung verfügbarer Zeitfenster eignet sich nicht für 

mehrere Benutzer. In [1] wird eine VBA-Funktion und präsentiert, die auf separaten, freigegebenen 

Kalendern aller einbezogenen Mitarbeiter beruht. Die Funktionalität dieser Funktion wird dort wie 

folgt beschrieben (übersetzt aus dem Englischen vom Verf. mit ‘Google Übersetzer’): 

“Mit dieser Funktion können Sie verfügbare Zeitfenster einer bestimmten Dauer für alle 

Mitarbeiter in einer Liste suchen. Sie übergeben das Datum und ein Array mit den 

Mitarbeiternamen an die Funktion und suchen nach freien Zeitfenstern mit einer Dauer von 

mindestens 2 Stunden zwischen einer festgelegten Start- und Endzeit an diesem Tag. Die 

Dauer und die Start-/Endzeiten sind Konstanten. Sie können diese Werte also entsprechend 

ändern. Die Funktion gibt eine Zeichenfolge zurück, die dann zum Auffüllen eines Listenfelds 

verwendet wird. Hier werden alle verfügbaren Zeitrahmen aufgelistet, und der Benutzer kann 

sich dann dafür entscheiden, einen Termin in einem dieser Zeitrahmen zu erstellen.” 

 

 

                                                             
4 Alle VBA-Prozeduren (ausgenommen Ereignisprozeduren) sind standardmäßig 'Public', also öffentlich. Wenn 
zu Beginn einer Prozedur das Schlüsselwort 'Private' deklariert wird, können nur die im selben VBA-Modul 
gespeicherten Prozeduren darauf zugreifen. 

Option Explicit 1 
' Globale Variable: Ein nullbasiertes, eindimensionales Datenfeld 2 
Dim varTm As Variant 3 
 4 
Private Sub UserForm_Initialize() 5 
   With Userform2 6 
      ' Voreinstellungen 7 
      .txtDate.Value = CDate("22. Okt. 2019")   ' Datum 8 
      .txtDuration.Value = 30                   ' normale Dauer eines Termins in Minuten 9 
      .txtStartTime.Value = "08:30"             ' Startzeit 10 
      .txtEndTime.Value = "19:00"               ' Endzeit ( > Startzeit) 11 
      .cmdCreate.Enabled = False                ' Befehlsschaltfläche deaktivieren 12 
      With .cboTimeSlots                        ' Kombinationsfeld f. Zeitfenster 13 
         .List = Array() 14 
         .Enabled = False 15 
         .BackColor = RGB(211, 211, 211) 16 
      End With 17 
      With .txtTopic                            ' Betreff 18 
         .Value = "" 19 
         .Enabled = False 20 
         .BackColor = RGB(211, 211, 211) 21 
      End With 22 
      With .txtWhere                            ' Ort 23 
         .Value = "" 24 
         .Enabled = False 25 
         .BackColor = RGB(211, 211, 211) 26 
      End With 27 
      With .txtBegin                            ' Beginn 28 
         .Value = "" 29 
         .Enabled = False 30 
         .BackColor = RGB(211, 211, 211) 31 
      End With 32 
      With .txtLength                           ' Dauer in Minuten 33 
         .Value = "" 34 
         .Enabled = False 35 
         .BackColor = RGB(211, 211, 211) 36 
      End With 37 
      With .chkRemind                           ' Erinnerung (Ja/Nein) 38 
         .Enabled = False 39 
         .BackColor = RGB(211, 211, 211) 40 
      End With 41 
      With .txtComment                          ' Textkörper 42 
         .Value = "" 43 
         .Enabled = False 44 
         .BackColor = RGB(211, 211, 211) 45 
      End With 46 
   End With 47 
End Sub 48 
 49 
Private Function FindAvailableTimeSlots(dtmAppt As Date, iDuration As Integer, tmFirstAppt As 50 
Date, tmLastAppt As Date) As String 51 
   ' Aufgabe:     Am übergebenen Tag (dtmDate) alle verfügbaren Zeitfenster zwischen den schon 52 
vorhandenen Terminen finden. 53 
   ' Argumente:   dtmAppt        Das zu durchsuchende Datum im Outlook-Kalender 54 
   '              iDuration      Normale Dauer eines Termine in Minuten 55 
   '              tmFirstAppt    Beginn des ersten Termins 56 
   '              tmLastAppt     Ende des letzten Termins 57 
   ' Annahmen:    Das freie Zeitfenster muss ... 58 
   '                 * größer oder gleich der normalen Dauer eines Termins sein. 59 
   '                    * zwischen dem normalen Beginn- und Ende der Arbeitszeit eines Tages liegen. 60 
   Const conFctName     As String = "FindAvailableTimeSlots" ' Name der Funktionsprozedur 61 
   Dim objOL            As New Outlook.Application ' OutLook-Anwendung 62 
   Dim olNS             As Outlook.NameSpace       ' Basis aller Ordner d. jeweiligen 63 
NameSpace 64 
   Dim olFldr           As Outlook.MAPIFolder      ' Verzeichnis 65 
   Dim olItems          As Outlook.items           ' Sammlung von Terminen 66 
   Dim olAppt           As Object                  ' einzelner Termin 67 
   Dim dtmNext          As Date                    ' Nächster verfügbarer Termin 68 
   Dim intDuration      As Integer                 ' Dauer der freien Lücke zwischen Terminen 69 
   Dim strFullDay       As String                  ' Voller Tag 70 
   Dim strList          As String                  ' Liste der verfügbaren Zeitfenster 71 
   On Error GoTo Error_Point 72 
   ' Anfangswert setzen 73 
   strList = "" 74 
   ' Den vollen Zeitraum des ausgewählten Tages bestimmen 75 
   strFullDay = "[Start] >= '" & dtmAppt & "' and " & "[Start] < '" & dtmAppt & " 11:59 pm'" 76 
   Set olNS = objOL.GetNamespace("MAPI") 77 
   On Error GoTo Error_Point 78 

   Set olFldr = olNS.GetDefaultFolder(olFolderCalendar) 79 
   Set olItems = olFldr.items 80 
   dtmNext = tmFirstAppt 81 
   ' Termine sortieren (erforderlich wg. "IncludeRecurrences", s. u.) 82 
   olItems.Sort "[Start]" 83 
   ' Wiederkehrende Termine einbeziehen 84 
   olItems.IncludeRecurrences = True 85 
   ' Die Termine so sortieren, dass nur alle Termine des relevanten Tages berücksichtigt werden. 86 
   Set olItems = olItems.Restrict(strFullDay) 87 
   ' Die Termine erneut sortieren , um wiederkehrende Termine in die richtige Reihenfolge zu bringen. 88 
   olItems.Sort "[Start]" 89 
   With olItems 90 
      ' Jeweils Betreff, Starttermin und Dauer ermitteln 91 
      Set olAppt = .GetFirst 92 
      Do While TypeName(olAppt) <> "Nothing" 93 
         ' Verfügbare Zeitfenster finden 94 
         Select Case DateValue(dtmAppt) ' Zeichenfolge in einen gültigen Datumswert umwandeln. 95 
            Case DateValue(Format(olAppt.Start, "dd.mm.yyyy")) 96 
               If Format(dtmNext, "Hh:Nn") < Format(olAppt.Start, "Hh:Nn") Then 97 
                   ' Die zeitliche Lücke bestimmmen bevor der nächste Termin beginnt. 98 
                  If Format(olAppt.Start, "Hh:Nn") < Format(tmLastAppt, "Hh:Nn") Then 99 
                      intDuration = DateDiff("n", dtmNext, Format(olAppt.Start, "Hh:Nn")) 100 
                  Else 101 
                      intDuration = DateDiff("n", dtmNext, Format(tmLastAppt, "Hh:Nn")) 102 
                  End If 103 
                  ' Passt ein Termin in das Zeitfenster? 104 
                  If intDuration >= iDuration Then 105 
                     strList = strList & Format(dtmNext, "hh:mm") & "|" _ 106 
                                          & Format(DateAdd("n",intDuration,dtmNext), "hh:mm") & ";" 107 
                  End If 108 
               End If 109 
               ' Das erste verfügbare Zeitfenster nach dem Termin finden. 110 
               dtmNext = DateAdd("n", olAppt.Duration + intDuration, dtmNext) 111 
               ' Nicht über den letztmöglichen Termin hinausgehen! 112 
               If dtmNext > tmLastAppt Then 113 
                  Exit Do 114 
               End If 115 
         End Select 116 
         intDuration = 0 117 
         Set olAppt = .GetNext 118 
      Loop 119 
   End With 120 
   ' Den Rest des Tages bestimmen. 121 
   intDuration = DateDiff("n", dtmNext, Format(tmLastAppt, "Hh:Nn")) 122 
   If intDuration >= iDuration Then 123 
      strList = strList & Format(dtmNext, "hh:mm") & _ 124 
                    "|" & Format(DateAdd("n", intDuration, dtmNext), "hh:mm") & ";" 125 
   End If 126 
   FindAvailableTimeSlots = strList 127 
   Me.cmdFind.Enabled = False 128 
Exit_Point: 129 
   On Error Resume Next 130 
   Set olAppt = Nothing 131 
   Set olItems = Nothing 132 
   Set olFldr = Nothing 133 
   Set olNS = Nothing 134 
   Set objOL = Nothing 135 
   Exit Function 136 
Error_Point: 137 
   MsgBox Err.Number & ": " & conFctName & vbCrLf & Err.Description 138 
   Resume Exit_Point 139 
End Function 140 
Private Function GetBusyStatus(intBusyStatus As Integer) As String 141 
   ' Verfügbarkeit ermitteln. 142 
   Select Case intBusyStatus 143 
      Case olFree 144 
        GetBusyStatus = "verfügbar" 145 
      Case olTentative 146 
        GetBusyStatus = "mit Vorbehalt" 147 
      Case olBusy 148 
        GetBusyStatus = "beschäftigt" 149 
      Case olOutOfOffice 150 
        GetBusyStatus = "abwesend" 151 
      Case Else 152 
        GetBusyStatus = "unbekannt (" & intBusyStatus & ")" 153 
    End Select 154 
End Function 155 
 156 

Private Function Check_Integer(Tbox As MSForms.TextBox) As Boolean 157 
   ' Minuten prüfen. 158 
   Dim strMsg  As String 159 
   With Tbox 160 
      If .Text = "" Then 161 
         strMsg = "Normale Dauer ist Pflichtfeld!" 162 
      ElseIf Not IsNumeric(.Text) Then 163 
         strMsg = "Eingabe ist nicht nummerisch!" 164 
      ElseIf CInt(.Text) <> .Text Then 165 
          strMsg = "Ganze Minuten sind erforderlich!" 166 
      End If 167 
   End With 168 
   If Len(strMsg) > 0 Then 169 
      MsgBox strMsg, vbExclamation, "Eingabefehler" 170 
      Check_Integer = True '  --> springt wieder in das Eingabefeld zurück 171 
   End If 172 
End Function 173 
Private Function Check_Time(Tbox As MSForms.TextBox) As Boolean 174 
   ' Zeitangabe prüfen. 175 
   Dim strMsg  As String 176 
   With Tbox 177 
      If .Text = vbNullString Then 178 
         strMsg = "Zeitabgabe ist Pflichtfeld!" 179 
      ElseIf Len(Trim(.Text)) <> 5 Then 180 
         strMsg = "Länge der Zeitabgabe ungleich 5!" 181 
      ElseIf InStr(3, .Text, ":", vbTextCompare) = 0 Then 182 
         strMsg = "Zeitabgabe enthält keinen Doppelpunkt!" 183 
      End If 184 
   End With 185 
   If Len(strMsg) > 0 Then 186 
      MsgBox strMsg, vbExclamation, "Eingabefehler" 187 
      Check_Time = True ' --> springt wieder in das Eingabefeld zurück 188 
   End If 189 
End Function 190 
Private Sub cmdFind_Click() 191 
   ' Verfügbare Zeitfenster bestimmen 192 
   Dim strResult As String 193 
   strResult = FindAvailableTimeSlots(Me.txtDate, Me.txtDuration, Me.txtStartTime, Me.txtEndTime) 194 
   Dim varResult As Variant 195 
   varResult = Split(strResult, ";") 196 
   If IsArray(varResult) Then 197 
      With Me.cboTimeSlots 198 
         .BackColor = RGB(255, 255, 255) 199 
         .Enabled = True 200 
         .List = varResult 201 
         .ListIndex = -1 202 
         .ListIndex = 0                         ' 1. Eintrag d. Kombinationsfelds anzeigen 203 
      End With 204 
      With Me 205 
         .Hide 206 
         With .txtDate                          ' Datum 207 
            .Enabled = False 208 
            .BackColor = RGB(211, 211, 211) 209 
         End With 210 
         With .txtDuration                      ' Dauer 211 
            .Enabled = False 212 
            .BackColor = RGB(211, 211, 211) 213 
         End With 214 
         With .txtStartTime                     ' Beginn 215 
            .Enabled = False 216 
            .BackColor = RGB(211, 211, 211) 217 
         End With 218 
         With .txtEndTime                       ' Ende 219 
            .Enabled = False 220 
            .BackColor = RGB(211, 211, 211) 221 
         End With 222 
         .Show 223 
      End With 224 
   End If 225 
   Unload Me 226 
End Sub 227 
  228 

Private Sub cmdCreate_Click() 229 
   Dim newAppt As Outlook.AppointmentItem 230 
   Set newAppt = Application.CreateItem(olAppointmentItem) 231 
   With newAppt 232 
       .Subject = Me.txtTopic.Value                   ' Betreff 233 
       .Location = Me.txtWhere.Value                  ' Ort 234 
       .AllDayEvent = False                           ' Kein ganztägiger Termin 235 
       .BusyStatus = olBusy                           ' Termim besetzt 236 
       .Categories = "Gelbe Kategorie"                ' Kategorie 237 
       .Start = Me.txtDate & " " & Left(varTm(0), 5)  ' Beginn 238 
       .ReminderSet = Me.chkRemind                    ' Erinnerung (Ja/Nein) 239 
       .Duration = Me.txtDuration.Value               ' Dauer in Minuten 240 
       .Body = Me.txtComment                          ' Kommentar (unformatierter Text) 241 
       .Save                                          ' Termin speichern 242 
   End With 243 
   With Me 244 
      .cmdCreate.Enabled = False 245 
      .cmdFind.Enabled = False 246 
   End With 247 
   Dim strMsg As String 248 
   With newAppt 249 
      strMsg = "Einmaliger Termin erfolgreich generiert!" & vbCrLf & _ 250 
               "-------------------------------------------" & vbCrLf & _ 251 
               "Betreff: " & .Subject & vbCrLf & _ 252 
               "Ort: " & .Location & vbCrLf & _ 253 
               "Beginn: " & .Start & vbCrLf & _ 254 
               "Ende: " & .End & vbCrLf & _ 255 
               "Dauer: " & .Duration & " Minuten" & vbCrLf & _ 256 
               "Ganztägiger Termin: " & .AllDayEvent & vbCrLf & _ 257 
               "Beschäftigungsstatus: " & GetBusyStatus(.BusyStatus) & vbCrLf & _ 258 
               "Kategorie: " & .Categories & vbCrLf & _ 259 
               "Erinnerung: " & .ReminderSet & vbCrLf & _ 260 
               "Textkörper: " & .Body & vbCrLf 261 
   End With 262 
   MsgBox strMsg, vbInformation + vbOKOnly, "Termin erstellen" 263 
   Set newAppt = Nothing 264 
End Sub 265 
Private Sub cmdCancel_Click() 266 
   Me.txtTopic = ""        ' Betreff 267 
   Me.txtWhere = ""        ' Ort 268 
   Me.txtBegin = ""        ' Beginn 269 
   Me.txtLength = ""       ' Dauer 270 
   Me.chkRemind = False    ' Erinnerung 271 
   Me.txtComment = ""      ' Kommentar 272 
   Unload Me               ' Benutzerformular aus Speicher entfernen 273 
End Sub 274 
Private Sub UserForm_QueryClose(Cancel As Integer, CloseMode As Integer) 275 
   ' Verhindern, dass das Benutzerformular geschlossen wird, wenn auf das obere rechte X geklickt wird. 276 
   If CloseMode = vbFormControlMenu Then 277 
      MsgBox "Bitte das Formular mit der Schaltfläche 'Abbrechen' schließen!" 278 
      Cancel = True 279 
   End If 280 
End Sub 281 
Private Sub txtDate_Exit(ByVal Cancel As MSForms.ReturnBoolean) 282 
   With Me.txtDate 283 
      If Not IsDate(.Text) And Not .Text = vbNullString Then 284 
         ' .Text = "" 285 
         MsgBox "Bitte ein gültiges Datum eingeben!", vbExclamation, "Eingabefehler" 286 
         Cancel = True ' ---> springt wieder in das Eingabefeld zurück 287 
      Else 288 
         On Error Resume Next 289 
         .Text = CDate(.Text) 290 
      End If 291 
   End With 292 
End Sub 293 
Private Sub txtStartTime_Exit(ByVal Cancel As MSForms.ReturnBoolean) 294 
   Cancel = Check_Time(txtStartTime) ' True --> springt wieder in das Eingabefeld zurück 295 
End Sub 296 
Private Sub txtEndTime_Exit(ByVal Cancel As MSForms.ReturnBoolean) 297 
   Cancel = Check_Time(txtEndTime) ' True --> springt wieder in das Eingabefeld zurück 298 
End Sub 299 
Private Sub txtTopic_Exit(ByVal Cancel As MSForms.ReturnBoolean) 300 
   If Len(Trim(Me.txtTopic.Value)) = 0 Then 301 
      MsgBox "Betreff ist Pflichteld!", vbExclamation, "Eingabefehler" 302 
      Cancel = True 303 
   End If 304 
End Sub 305 
Private Sub txtWhere_Exit(ByVal Cancel As MSForms.ReturnBoolean) 306 

   If Len(Trim(Me.txtWhere.Value)) = 0 Then 307 
      MsgBox "Ort ist Pflichteld!", vbExclamation, "Eingabefehler" 308 
      Cancel = True 309 
   End If 310 
End Sub 311 
Private Sub txtDuration_Exit(ByVal Cancel As MSForms.ReturnBoolean) 312 
   Cancel = Check_Integer(txtDuration) 313 
End Sub 314 
Private Sub txtEndTime_AfterUpdate() 315 
   On Error Resume Next 316 
   If CDate(Me.txtEndTime.Text) <= CDate(Me.txtStartTime.Text) Then 317 
      MsgBox "Ende vor/gleich Beginn!", vbExclamation, "Eingabefehler" 318 
   End If 319 
End Sub 320 
Private Sub cboTimeSlots_Change() 321 
   ' Zeitfenster ändert sich. 322 
   If Me.cboTimeSlots.ListIndex = -1 Then 323 
      MsgBox "Nichts ausgewählt!", vbExclamation, "Zeitfenster" 324 
   Else 325 
      ' Untergeordnete Zeichenfolgen im Kombinationsfeld f. Zeitfenster bestimmen. 326 
      varTm = Split(Me.cboTimeSlots.Value, "|") 327 
      ' Fomularfelder aufbereiten 328 
      With Me 329 
         .cmdCreate.Enabled = True 330 
         With .txtTopic 331 
            .Enabled = True 332 
            .BackColor = RGB(255, 255, 255) 333 
            .Value = "Neuer Termin" 334 
         End With 335 
         With .txtWhere 336 
            .Enabled = True 337 
            .BackColor = RGB(255, 255, 255) 338 
            .Value = "Praxis" 339 
         End With 340 
         With .txtComment 341 
            .Enabled = True 342 
            .BackColor = RGB(255, 255, 255) 343 
            .Value = "Kommentar" 344 
         End With 345 
         With .chkRemind 346 
            .Enabled = True 347 
            .BackColor = RGB(255, 255, 255) 348 
            .Value = False 349 
         End With 350 
         .txtBegin.Value = Me.txtDate & " " & Left(varTm(0), 5) 351 
         .txtLength.Value = .txtDuration.Value 352 
      End With 353 
   End If 354 
End Sub 355 

Listing 1: Gesamter VBA-Quellcode der Anwendung 

